
Motivating example
● “read a string, if it is ‘ping’ then print ‘pong’, 

otherwise do nothing”

pingPongM :: IO ()
pingPongM = 
  getLine >>= \s -> 
    if s == "ping" then putStrLn "pong" else pure ()



What if we want to analyse it?

● Sometimes it’s useful to be able to ask “what are all the 
effects this computation might have?”

● We could use this to
○ pre-allocate resources
○ speculate execution, parallelism
○ (examples coming later)

pingPongM :: IO ()
pingPongM = 
  getLine >>= \s -> 
    if s == "ping" then putStrLn "pong" else pure ()



But we cannot do that here!
pingPongM :: IO ()
pingPongM = 
  getLine >>= \s -> 
    if s == "ping" then putStrLn "pong" else pure ()

Only known at runtime



In general, Monad makes this impossible
class Monad f where
  return :: a -> f a
  (>>=)  :: f a -> (a -> f b) -> f b

● We cannot know a until we have peformed f a
● So we cannot analyse the computation to find all its 

(potential) effects, we can only run it.



But let’s take a simpler example
whenM :: Monad m => m Bool -> m () -> m ()

first execute this...



But let’s take a simpler example
whenM :: Monad m => m Bool -> m () -> m ()

first execute this...
if it returned True, 

execute this, 
otherwise don’t



Rewrite our example using whenM

We will need fmap:

Now, to get IO Bool:

whenM :: Monad m => m Bool -> m () -> m ()

class Functor f where
  fmap :: (a -> b) -> f a -> f b

fmap (== “ping”) getLine :: IO Bool



Rewrite our example using whenM

pingPongM :: IO ()
pingPongM = 
  whenM (fmap (== “ping”) getLine) (putStrLn "pong")

whenM :: Monad m => m Bool -> m () -> m ()

pingPongM :: IO ()
pingPongM = 
  getLine >>= \s -> 
    if s == "ping" then putStrLn "pong" else pure ()



But why is this better?
● Look at the definition of whenM:

whenM :: Monad m => m Bool -> m () -> m ()
whenM x y = x >>= \b -> if b then y else return ()



But why is this better?
● Look at the definition of whenM:

● We have some hope of statically analysing this code, because 
we can enumerate all the possibilities for b

whenM :: Monad m => m Bool -> m () -> m ()
whenM x y = x >>= \b -> if b then y else return ()

Still a runtime value, but 
it only has two possible 
values



But why is this better?
● Look at the definition of whenM:

● We have some hope of statically analysing this code, because 
we can enumerate all the possibilities for b

● But we can’t do it in this form, using >>=

whenM :: Monad m => m Bool -> m () -> m ()
whenM x y = x >>= \b -> if b then y else return ()

Still a runtime value, but 
it only has two possible 
values



But wait...
● Don’t we already have an abstraction that...

○ is weaker than Monad
○ admits static analysis



Applicative Functors: 2007, Nottingham/London



Applicative Functors
class Applicative f where
  pure  :: a -> f a
  (<*>) :: f (a -> b) -> f a -> f b

✔ We can execute computations f (a -> b) and f a in parallel (if we like).

✔ All effects are statically visible and can be examined before execution.

✘ Computations must be independent, hence no conditional execution.



Towards a new intermediate abstraction

Applicative 
functors

??? Monads



Towards a new intermediate abstraction

Applicative 
functors

??? Monads

Independent effects & parallelism ✔ ✘

(×) :: f a -> f b -> f (a,b)



Towards a new intermediate abstraction

Applicative 
functors

??? Monads

Independent effects & parallelism ✔ ✘

Static visibility & analysis of effects ✔ ✘

  getPure    :: f a -> Maybe a
  getEffects :: f a -> [f ()]



Towards a new intermediate abstraction

Applicative 
functors

??? Monads

Independent effects & parallelism ✔ ✘

Static visibility & analysis of effects ✔ ✘

Dynamic generation of effects ✘ ✔

greeting = getLine >>= \name -> putStrLn ("Hello " ++ name)



Towards a new intermediate abstraction

Applicative 
functors

??? Monads

Independent effects & parallelism ✔ ✘

Static visibility & analysis of effects ✔ ✘

Dynamic generation of effects ✘ ✔

Conditional execution of effects ✘ ✔

pingPongM = whenM (fmap (=="ping") getLine) (putStrLn "pong")



Towards an intermediate abstraction

Applicative 
functors

??? Monads

Independent effects & parallelism ✔ ✘

Static visibility & analysis of effects ✔ ✘

Dynamic generation of effects ✘ ✔

Conditional execution of effects ✘ ✔

Speculative execution of effects ✘ ✘

Ad-hoc speculative execution combinators from the Haxl library:

pAnd :: f Bool -> f Bool -> f Bool
pOr  :: f Bool -> f Bool -> f Bool



Towards an intermediate abstraction

Applicative 
functors

Selective 
functors

Monads

Independent effects & parallelism ✔ ✘

Static visibility & analysis of effects ✔ ✘

Dynamic generation of effects ✘ ✔

Conditional execution of effects ✘ ✔

Speculative execution of effects ✘ ✘



Towards an intermediate abstraction

Applicative 
functors

Selective 
functors

Monads

Independent effects & parallelism ✔ ✔ ✘

Static visibility & analysis of effects ✔ ✔ ✘

Dynamic generation of effects ✘ ✘ ✔

Conditional execution of effects ✘ ✔ ✔

Speculative execution of effects ✘ ✔ ✘



Selective Applicative Functors
● Goal: an abstraction that allows

○ static analysis, parallelism, speculative execution
○ conditional effects



Selective Applicative Functors
● Goal: an abstraction that allows

○ static analysis, parallelism, speculative execution
○ conditional effects

class Applicative f => Selective f where
  select :: f (Either a b) -> f (a -> b) -> f b

The first computation is used to select what happens next:
•Left a: you must execute the second computation to produce a b;
•Right b: you may skip the second computation and return the b.



Selective Applicative Functors

✔ We can speculatively execute both computations in parallel (if we 
like).

✔ All effects are statically visible and can be examined before 
execution.

✔ A limited form of dependence, sufficient for conditional execution.

class Applicative f => Selective f where
  select :: f (Either a b) -> f (a -> b) -> f b



Why this particular formulation?

● Parametricity tell us what select can do
○ whenM can be implemented wrongly (unlessM)

class Applicative f => Selective f where
  select :: f (Either a b) -> f (a -> b) -> f b



But we love operators, so
(<*?) :: Selective f => f (Either a b) -> f (a -> b) -> f b
(<*?) = select
 



Example
pingPongS :: IO ()
pingPongS = whenS (fmap (=="ping") getLine) (putStrLn "pong")

whenS :: Selective f => f Bool -> f () -> f ()
whenS x y = selector <*? effect
  where
selector :: f (Either () ())
selector = bool (Right ()) (Left ()) <$> x

effect :: f (() -> ())
effect = const <$> y



What interesting combinators can we build?

Define branch in terms of select...

branch :: Selective f => f (Either a b) -> f (a -> c) -> f (b -> c) -> f c

select :: Selective f => f (Either p q) -> f (p -> q) -> f q



What interesting combinators can we build?

Define branch in terms of select...

branch :: Selective f => f (Either a b) -> f (a -> c) -> f (b -> c) -> f c

select :: Selective f => f (Either p q) -> f (p -> q) -> f q

branch x l r = fmap (fmap Left) x <*? fmap (fmap Right) l <*? r



More combinators
ifS :: Selective f => f Bool -> f a -> f a -> f a

ifS x t e = branch (bool (Right ()) (Left ()) <$> x) (const <$> t) (const <$> e)

(<||>) :: Selective f => f Bool -> f Bool -> f Bool

a <||> b = ifS a (pure True) b

(<&&>) :: Selective f => f Bool -> f Bool -> f Bool

a <&&> b = ifS a b (pure False)

anyS :: Selective f => (a -> f Bool) -> [a] -> f Bool

anyS p = foldr ((<||>) . p) (pure False)

allS :: Selective f => (a -> f Bool) -> [a] -> f Bool

allS p = foldr ((<&&>) . p) (pure True)



Every Monad is Selective
● selectM :: Monad m => m (Either a b) -> m (a -> b) -> m b

selectM x y = x >>= \e -> 
  case e of 
    Left  a -> ($a) <$> y
    Right b -> return b



Every Monad is Selective

● In fact, select = selectM is the definition of the 
semantics of select for a Monad.

○ (rather like <*> = ap defines the semantics of Applicative 

for a Monad)

selectM :: Monad m => m (Either a b) -> m (a -> b) -> m b
selectM x y = x >>= \e -> 
  case e of 
    Left  a -> ($a) <$> y
    Right b -> return b



Every Monad is Selective

● Some Monads may choose to implement select 
more efficiently

○ e.g. Haxl uses parallelism for <*>, speculation for 
select

selectM :: Monad m => m (Either a b) -> m (a -> b) -> m b
selectM x y = x >>= \e -> 
  case e of 
    Left  a -> ($a) <$> y
    Right b -> return b



Every Applicative is Selective
selectA :: Applicative f => f (Either a b) -> f (a -> b) -> f b
selectA x y = (\e f -> either f id e) <$> x <*> y

● This is a valid implementation of select, 
○ but may not be the only one.

● Summary:
○ select = selectM →  conditional effects
○ select = selectA →  unconditional effects

Always executes y



Data validation example
data Validation e a = Failure e | Success a 

instance Semigroup e => Applicative (Validation e) where
pure = Success
Failure e1 <*> Failure e2 = Failure (e1 <> e2)
Failure e1 <*> Success _  = Failure e1
Success _  <*> Failure e2 = Failure e2
Success f  <*> Success a  = Success (f a)

The idea is that we can 
traverse a structure and 

report multiple errors



Data validation example
data Validation e a = Failure e | Success a 

instance Semigroup e => Selective (Validation e) where
select (Success (Right b)) _ = Success b
select (Success (Left  a)) f = ($a) <$> f
select (Failure e       ) _ = Failure e

Accumulates errors in 
both computations



Data validation example

● Neither selectA nor selectM
● Cannot be a Monad!

data Validation e a = Failure e | Success a 

instance Semigroup e => Selective (Validation e) where
select (Success (Right b)) _ = Success b
select (Success (Left  a)) f = ($a) <$> f
select (Failure e       ) _ = Failure e

Discard errors on the 
right if the condition 

failed



mkAddress
  :: Selective f
  => f Street
  -> f City
  -> f PostCode
  -> f Country
  -> f Address

mkAddress street city postcode country =
  Address
   <$> street
   <*> city
   <*> ifS (hasPostCode <$> country) (Just <$> postcode) (pure Nothing)
   <*> country
  



Laws
● There are identity, distributive and associative laws
● Non-laws:
○ pure (Right x) <*? y == pure x

○ pure (Left x)  <*? y == ($x) <$> y

○ these would rule out over-approximation and 
under-approximation of effects

● But: Monads must satisfy select = selectM



Selective and Haxl



What is Haxl?
● Solves the following problem:

○ I want to write code that works with remote data
○ I want data-fetching to happen in parallel where possible
○ automatically, without me having to do anything

● In use at scale at Facebook for writing anti-abuse code



● In Haxl:
○ Applicative = parallel
○ Monad = sequential

● e.g. mapM fetch things will fetch things in parallel



Conditionals
● We found short-cutting “and” and “or” useful:

● Particularly in cases like

(.||), (.&&) :: Haxl Bool -> Haxl Bool -> Haxl Bool

a .&& b = do
  x <- a
  if x then b else return False

if simpleCondition .&& complexCondition then .. else ..



● But sometimes it’s not easy to know the best ordering

● … especially when the number of conditions is large, 
and/or changes often

● We could do it in parallel:

● But this leaves some performance on the table:
○ if either condition returns False early, we don’t need 

to finish evaluating the other one.

   complexCondition .&& otherComplexCondition

   and [complexCondition, otherComplexCondition]



● These are semantically the same as (.&&), (.||)
○ but evaluate both arguments in parallel
○ and bail out early if the answer is known

Parallel boolean operators
pAnd, pOr :: Haxl Bool -> Haxl Bool -> Haxl Bool



● But if we define

● Now

● And the rest of the selective combinators will now be 
speculative/parallel.

pAnd = (<&&>)
pOr = (<||>)

instance Selective Haxl where
  select = ...



But there’s a subtle problem...
● select will always execute its first argument to 

completion
● whereas we want pAnd to abort the first argument if the 

second argument returns False
○ e.g. (someFetch >>= x) `pAnd` return False
○ should never execute x



Select is not precisely what we want
● But we can define a symmetric alternative:

● Solution: Add biselect as a method in Selective
● Instances can override biselect if they want

biselect 
  :: Selective f
  => f (Either a b)
  -> f (Either a c)
  -> f (Either a (b,c))



Generalisation
We have:

Alternatively:
ifS :: Selective f => f Bool -> f a -> f a -> f a

bindBool :: Selective f => f Bool -> (Bool -> f a) -> f a



Generalisation
We have:

Alternatively:

Moreover:

ifS :: Selective f => f Bool -> f a -> f a -> f a

bindBool :: Selective f => f Bool -> (Bool -> f a) -> f a

bindS
 :: (Selective f, Bounded a, Enum a, Eq a) 
 => f a -> (a -> f b) -> f b

Look familiar?



bindS

● Implementation in terms of select could 
sequentially check all the possible values of a

● But for a monad, bindS = (>>=)
○ suggests that bindS should be a method

bindS
 :: (Selective f, Bounded a, Enum a, Eq a) 
 => f a -> (a -> f b) -> f b



More applications
● Build systems: 

○ extract all build dependencies before execution, with 
conditional execution

● Modelling processor instructions:
○ Categorising instructions: Functor (e.g. increment), Applicative 

(arithmetic), Selective (branching), Monad (indirect memory 
access)

● Parsing combinators:
○ Use Selective instead of Alternative to avoid backtracking



Conclusions
● Selective identifies a useful point in the design space 

between Applicative and Monad
● Combines the benefits of Applicative (static analysis, 

parallelism, speculation) with limited conditional 
support



Example: a blog engine

I want to fetch all the content of all the posts:

● Just use standard monadic combinators
● mapM getPostContent should happen in parallel

getPostIds     :: Haxl [PostId]
getPostContent :: PostId -> Haxl PostContent

getAllPostsContent :: Haxl [PostContent]
getAllPostsContent = getPostIds >>= mapM getPostContent



Batching
Indeed, not just parallel, but batching multiple 
requests where possible:

SELECT content FROM posts
  WHERE postid = id1

SELECT content FROM posts
  WHERE postid = id2

...

SELECT content FROM posts
  WHERE postid IN {id1, id2, ...}

Unbatched Batched



Implementation
data Result a
  = Done a
  | Blocked (Seq BlockedRequest) (Haxl a)

newtype Haxl a = Haxl { unHaxl :: IO (Result a) }

This is the 
result of a 

computation



Implementation
data Result a
  = Done a
  | Blocked (Seq BlockedRequest) (Haxl a)

newtype Haxl a = Haxl { unHaxl :: IO (Result a) }

Done indicates 
that we have 

finished



Implementation
data Result a
  = Done a
  | Blocked (Seq BlockedRequest) (Haxl a)

newtype Haxl a = Haxl { unHaxl :: IO (Result a) }

Blocked indicates that the 
computation requires this 

data.



Implementation
data Result a
  = Done a
  | Blocked (Seq BlockedRequest) (Haxl a)

newtype Haxl a = Haxl { unHaxl :: IO (Result a) }

Haxl is in IO, 
because we 

use IORefs to 
store results



instance Monad Haxl where
  return a = Haxl $ return (Done a)

  Haxl m >>= k = Haxl $ do
    r <- m
    case r of
      Done a       -> unHaxl (k a)
      Blocked br c -> return (Blocked br (c >>= k))

If m blocks with continuation c, the 
continuation for m >>= k is c >>= k



instance Applicative Haxl where
  pure = return

  Haxl f <*> Haxl x = Haxl $ do
    f' <- f
    x' <- x
    case (f',x') of
      (Done g,        Done y       ) -> return (Done (g y))
      (Done g,        Blocked br c ) -> return (Blocked br (g <$> c))
      (Blocked br c,  Done y       ) -> return (Blocked br (c <*> return y))
      (Blocked br1 c, Blocked br2 d) -> return (Blocked (br1 <> br2) (c <*> d))

Haxl works by having a special Applicative instance

● when we use <*> we get parallelism
● when we use >>= we get sequentiality



Direct implementation
pAnd :: Haxl Bool -> Haxl Bool -> Haxl Bool
pAnd (Haxl a) (Haxl b) = Haxl $ do
  x <- a
  case x of
    Done False -> return False
    Done True -> b
    Blocked bx cx -> do
      y <- a
      case y of
        Done False -> return False
        Done True -> return x
        Blocked by cy -> 
          Blocked (bx <> by) (cx `pAnd` cy)



● When we say this is “parallel”, what do we mean?
○ data-fetches are done in parallel where possible
○ if both sides get blocked, we do their fetches together
○ NOT that we do the computation in parallel



Using Selective
instance Selective Haxl where
select (Haxl x) (Haxl f) = Haxl $ do

    rx <- x
    case rx of
      Done (Right b) -> return (Done b)
       Done (Left  a) -> unHaxl (($a) <$> Haxl f)
      Blocked bx c   -> do
         rf <- f
         case rf of
           Done f  -> unHaxl (either f id <$> c)
            Blocked by d -> 
              return (Blocked (bx <> by) (select c d))



What interesting combinators can we build?

Define branch in terms of select...

branch :: Selective f => f (Either a b) -> f (a -> c) -> f (b -> c) -> f c

select :: Selective f => f (Either p q) -> f (p -> q) -> f q

branch x l r = select x l

Would make b == c



What interesting combinators can we build?

Define branch in terms of select...

branch :: Selective f => f (Either a b) -> f (a -> c) -> f (b -> c) -> f c

select :: Selective f => f (Either p q) -> f (p -> q) -> f q

branch x l r = 
  select 
    (fmap (either Left (Right . Left)) x) 
    (fmap (\f -> Right . f) l)

q = Either b c



What interesting combinators can we build?

Define branch in terms of select...

branch :: Selective f => f (Either a b) -> f (a -> c) -> f (b -> c) -> f c

select :: Selective f => f (Either p q) -> f (p -> q) -> f q

branch x l r = 
  select (
    select 
      (fmap (either Left (Right . Left)) x) 
      (fmap (\f -> Right . f) l)
    ) r

q = Either b c



What interesting combinators can we build?

Define branch in terms of select...

branch :: Selective f => f (Either a b) -> f (a -> c) -> f (b -> c) -> f c

select :: Selective f => f (Either p q) -> f (p -> q) -> f q

branch x l r = 
  select (
    select 
      (fmap (either Left (Right . Left)) x) fmap (fmap Left)
      (fmap (\f -> Right . f) l)
    ) r



What interesting combinators can we build?

Define branch in terms of select...

branch :: Selective f => f (Either a b) -> f (a -> c) -> f (b -> c) -> f c

select :: Selective f => f (Either p q) -> f (p -> q) -> f q

branch x l r = 
  select (
    select 
      (fmap (either Left (Right . Left)) x) fmap (fmap Left)
      (fmap (\f -> Right . f) l)  fmap (fmap Right)
    ) r


